


Součet buněk dle barvy (pozadí) patří k nejčastějším zadáním objevující se na internetových fórech. Dodnes není uspokojivě vyřešen z pohledu funkcí listu. Na druhou stranu budme rádi za přidanou funkčnost v automatickém filtru, jíž si ukážeme posléze. Ještě předtím ale pár slov k tématu obecně.

Při každé příležitosti zdůrazňuji:

- 1) Smysl má barvit pozadí buňky, barva písma (ať už definovaná ručně, vlastním formátem buňky nebo podmíněným formátováním) se velmi špatně zrakem rozlišuje, pokud se nejedná o základní barvu a není dostatečně kontrastní s pozadím buňky
- 2) Veškeré barvičkování končí černobílou laserovou tiskárnou, kdy se slijí barvy s podobnou „sytostí“ do jednoho odstínu šedé. V grafech vám pak nepomůže ani legenda. Ne nadarmo byli uživatelé nešťastní, když nakrátko zmizely možnosti výplní grafů typu šrafování apod. U spojnic doporučuji rozlišovat nejen barvu čáry, ale i její typ. Na listu by teoreticky pomohlo ohraničení buňky, ale takový způsob je prostě tfujtajblový.
- 3) I když máte možnost barevného tisku, odstín barvy na monitoru a po tisku se bude lišit. Podstatnější ovšem je, že nemalá část populace trpí barvoslepostí ([test](#)) a dalšími očními neduhy, nemluvě o klasickém problému vnímání barev (zelená/modrá apod.).
- 4) Excel 2003 a starší nemá žádný nástroj pro práci s buňkami na základě barvy kromě možností v dialogu Najít a nahradit.

## Dialog Najít a nahradit

Už v základech Excelu říkám „klepněte si na tlačítko Možnosti v dialogu Najít a nahradit“. Rozbalí se nabídka dalších nástrojů, mezi nimiž je i možnost vyhledávání buněk na základě formátu. Pokud se přepnete na záložku Nahradit, pak můžete snadno buňky i přebarvovat.



Dialog Najít a nahradit – hledání dle formátu

Podstatná je ve vztahu k dnešnímu tématu jedna technika. Jestliže zvolíte Najít vše a ve výpisu stisknete CTRL+A, vyberou se nejen všechny položky v seznamu, ale i odpovídající buňky na listu! Po zavření dialogu se tak můžete podívat do stavového řádku řekněme na součet.


Pozn. Nezapomínejte před ukončením práce s dialogem resetovat nastavení formátu (Volba Vymazat pole pro hledání formátu).

Tento způsob práce s barevnými buňkami jednoduše nemohl chybět, i když v daném případě nelze mluvit o nějaké efektivní práci.

## Automatický filtr

Klíčem k řešení úlohy za pomoci vestavěných nástrojů je skutečnost, že novodobé verze Excelu již umí filtrovat dle barvy pozadí (i písma), dokonce si poradí i s obarvením vzniklým podmíněným formátem. Nad takto přefiltrovaným sloupcem pak stačí nasadit funkci SUBTOTAL s potřebnou matematickou operací.

	A	B	C	D	E	F
1						
2		SUBTOTAL	3160	=SUBTOTAL(9;C5:C24)		
3						
4			BARVA			
5			1			
6						
7						
8			10			
9						
10						
11			3			
12			4			
13			20			
14						
15			2			
16			30			
17			50			
18			40			
19						
20			1000			
21						
22						
23			2000			
24						
25						



	A	B	C	D	E	F
1						
2		SUBTOTAL	150	=SUBTOTAL(9;C5:C24)		
3						
4			BARVA			
5						
6						
7						
8			10			
9						
10						
11						
12						
13			20			
14						
15						
16			30			
17			50			
18			40			
19						
20						
21						
22						
23						
24						
25						

### Automatický filtr - filtrování dle barvy

Pokud barevné buňky nejsou rozházené po celém listu, ale řešíme danou problematiku v rámci sloupců seznamu, pak je automatický filtr nejefektivnější metodou pro součet buněk dle barvy (a nejen součet, viz první parametr funkce SUBTOTAL).

## Vlastní funkce VBA

Jak bylo řečeno, před příchodem Excelu 2007 neexistoval žádný rozumný způsob řešení úlohy bez VBA. Starší verze jednoduše neuměly filtrovat buňky podle barvy. Uvažujeme-li o řešení ve Visual Basicu pro aplikace, pak je třeba si uvědomit, že změna barvy buňky nevyvolá událost Change

(změna obsahu buňky), ani Calculate (přepočítání listu). I když si tedy vytvoříme vlastní funkci listu (UDF) a přinutíme ji být tzv. Volatile, přepočte se až v momentě jinak vzniklé potřeby přepočtu listu, případně při ručně vynuceném přepočtu (např. klávesou F9). A bohužel, pod VBA je velmi obtížné zjišťovat barvu coby výsledek podmíněného formátování. Prakticky dodnes se uplatňuje způsob, s jakým přišel Chip Pearson (přečtěte si články [Color Functions In Excel](#) a [Conditional Formatting Colors](#)). V podstatě procházíme všechny větve podmíněných formátů a testujeme, zda-li je v danou chvíli podmínka platná. Musíme zohlednit, že barva podmíněného formátování má přednost před ručně definovanou barvou, navíc u podmíněných formátů záleží na pořadí a „stopce“. Zpracování takové funkčnosti jsem se s prominutím vyhnul, níže uvedená vlastní funkce pracuje pouze s ručně definovanými barvami. Chápu, že tím ztrácí na kráse a užité hodnotě, ale...

```

1 Function epfFUNKCEBARVA(Oblast As Range, RefBunka As Range, Optional Operace As _
2   String = "součet") As Double
3
4   Dim rngBunka As Range
5
6   Dim arrPoleHodnot()
7
8   Dim lngBarva As Long
9   Dim i As Long
10
11  Dim dblHodnota As Double
12
13  'funkce zareaguje na přepočítání listu (nikoliv na obarvení buňky)
14  Application.Volatile
15
16  'barva pozadí referenční buňky
17  lngBarva = RefBunka.Interior.Color
18
19  'přeskočení chyb
20  On Error Resume Next
21
22  'dimenzování pole
23  ReDim arrPoleHodnot(1 To Oblast.Cells.Count)
24
25  'pro každou buňku v oblasti
26  For Each rngBunka In Oblast
27
28    'shoduje se barva pozadí buňky s referenční barvou
29    'a obsahuje číselnou hodnotu (může být i datum...)?
30    If (rngBunka.Interior.Color = lngBarva) And (IsNumeric(rngBunka.Value)) _
31      Then
32
33      'počítadlo
34      i = i + 1
35
36      'přidání číselného obsahu do součtu
37      arrPoleHodnot(i) = rngBunka.Value
38
39      End If
40
41  Next rngBunka
42
43  Select Case LCase(Operace)
44
45    Case "počet"
46
47      epfFUNKCEBARVA = WorksheetFunction.Count(arrPoleHodnot)
48
49    Case "součet"
50
51      epfFUNKCEBARVA = WorksheetFunction.Sum(arrPoleHodnot)
52
53    Case "průměr"
54
55      epfFUNKCEBARVA = WorksheetFunction.Average(arrPoleHodnot)
56
57    Case "minimum"
58
59      epfFUNKCEBARVA = WorksheetFunction.Min(arrPoleHodnot)
60
61    Case "maximum"
62
63      epfFUNKCEBARVA = WorksheetFunction.Max(arrPoleHodnot)
64
65  End Select
66
67 End Function

```

Funkce pracuje pouze s pozadím buňky. Očekává dva povinné argumenty – sčítanou oblast a referenční buňku s barvou, a jeden nepovinný – operaci prováděnou na hodnotách odpovídajících buněk. Součet buněk dle barvy nemusí být jedinou potřebnou operací, proto byl přidán i počet číselných buněk, průměr, minimum a maximum. Není problém za pomoci funkcí listu (WorksheetFunction) doplnit další.

	A	B	C	D	E	F	G	H	I
1									
2		1		Referenční buňka					
3				Počet čísel	4	5			
4				Součet	10	150			
5		10		Průměr	2,5	30			
6				Minimum	1	10			
7				Maximum	4	50			
8		3							
9		4		<i>vlastní funkce VBA (viz ALT+F11 a modul modBarva)</i>					
10		20		<i>pro spolehlivé fungování funkce je nutný stisk F9 (přepočítání listu)</i>					
11				<i>funkce nezahrnuje obarvení vzniklé na základě podmíněného formátování</i>					
12		2							
13		30							
14		50							
15		40							
16									
17		1000							
18									
19									
20		2000							
21									
22									
23									

### Vlastní funkce pro práci s barevnými buňkami

Na závěr bych doplnil ještě jednu proceduru pro případ, kdy nám jde pouze o seznam buněk s uvedením počtu jejich výskytu v dané oblasti. Kód využívá objekt Dictionary a jeho vlastnosti ověřovat si duplicitní výskyt položky.

```

1 Sub Barvy()
2
3 'Tools / References / Microsoft Scripting Runtime
4
5 Dim objDic As New Dictionary
6
7 Dim rngBunka As Range
8 Dim rngOblast As Range
9
10 Dim strHlaska As String
11 Dim strTitulek As String
12
13 Dim lngBarva As Long
14 Dim i As Integer
15
16 Dim PoleKlice()
17 Dim PolePolozky()
18
19 'texty v dialogu
20 strHlaska = "Myší označte jednosloupcovou, spojitou oblast buněk."
21 strTitulek = "Zpracovávaná oblast"
22
23 'přechod na další řádek v případě, že se nepodaří
24 'přiřazení oblasti do objektové proměnné
25 On Error Resume Next
26
27 'vlastní pokus o přiřazení oblasti z dialogu do objektové proměnné
28 Set rngOblast = Application.InputBox(strHlaska, strTitulek, _
29 Selection.Address, , , , 8)
30
31 'opuštění procedury v případě chyby
32 If Err <> 0 Then Exit Sub
33
34 'pro každou buňku v oblasti
35 For Each rngBunka In rngOblast
36
37 On Error Resume Next
38
39 'přiřazení barvy buňky do proměnné
40 lngBarva = rngBunka.Interior.Color
41
42 'existuje záznam o barvě v knihovně?
43 If objDic(lngBarva).Exists = True Then
44
45 'ano, navýšit informaci o počtu výskytů
46 objDic(lngBarva) = objDic(lngBarva) + 1
47
48 Else
49
50 'ne, přidat záznam o barvě a informaci o prvním výskytu
51 objDic.Add lngBarva, 1
52
53 End If
54
55 Next rngBunka
56
57 'přenos klíčů a hodnot z objektu knihovny do polí
58 PoleKlice = objDic.Keys
59 PolePolozky = objDic.Items
60
61 'texty v dialogu
62 strHlaska = "Myší označte počátek vložení výsledku."
63 strTitulek = "Cílová oblast"
64
65 'přechod na další řádek v případě, že se nepodaří
66 'přiřazení oblasti do objektové proměnné
67 On Error Resume Next
68
69 'vlastní pokus o přiřazení oblasti z dialogu do objektové proměnné
70 Set rngOblast = Application.InputBox(strHlaska, strTitulek, , , , , 8)
71
72 'opuštění procedury v případě chyby
73 If Err <> 0 Then Exit Sub
74
75 'zamezení překreslování obrazovky
76 Application.ScreenUpdating = False
77
78 'pro každou buňku v cílové oblasti
79 For Each rngBunka In rngOblast.Cells(1).Resize(UBound(PoleKlice) + 1, 1)
80
81 'počítadlo
82 i = i + 1
83
84 'přiřazení barvy buňce z pole s klíči
85 rngBunka.Interior.Color = PoleKlice(i - 1)
86
87 'přiřazení hodnoty do buňky
88 rngBunka.Value = PolePolozky(i - 1)
89
90 Next rngBunka
91
92 'povolení překreslování obrazovky
93 Application.ScreenUpdating = True
94
95 End Sub

```

	A	B	C	D	E
1					
2		Orange		Orange	3
3		Yellow		Yellow	2
4		Green		Green	1
5		Grey		Grey	2
6		Yellow		Cyan	2
7		Cyan		Pink	3
8		Orange		Blue	1
9		Cyan		White	1
10		Pink		Green	3
11		Grey			
12		Orange			
13		Pink			
14		Blue			
15		White			
16		Pink			
17		Green			
18		Green			
19		Green			
20					
21					

Výčet barevných buněk

Příloha

[soucet\\_dle\\_barvy.zip](#)